Charge et décharge de condensateurs

1. Étudier un algorithme de résolution de ce problème d'électricité qui se présenterait de la façon suivante :

   — Définition des caractéristiques de composants.

   — Définition des lois de Kirchhoff.

   — Résolution des lois de Kirchhoff dans le but d'obtenir $u_1(t)$ et $u_2(t)$.

   Nous considérerons que l'ordinateur, pour notre bonheur, connaît parfaitement les mathématiques et, pour notre malheur, ne connaît rien à la physique.

   Écrire un programme en langage MAPLE qui réalise cet algorithme et aboutisse au tracé à l'écran des variations de $u_1(t)$ et $u_2(t)$ après en avoir donné l'expression littérale.

```
> restart:
  i1:=C*diff(u1(t),t);
  i2:=C*diff(u2(t),t);
  u3:=R*i2;
  loi_des_noeuds:=i1+i2=k;
  loi_des_mailles:=u1(t)=u3+u2(t);
  lois_de_Kirchhoff:=loi_des_noeuds,loi_des_mailles;
  tensions:=u1(t),u2(t);
  conditions_initiales:=u2(0)=0,u1(0)=0;
  dsolve({lois_de_Kirchhoff,conditions_initiales},{tensions});
  assign(%);
  u1:=unapply(u1(t),t);u2:=unapply(u2(t),t);
  R:=500:C:=1000e-6:T:=1:k:=50e-3:
  plot([u1,u2],0..T,thickness=3);
```

$$u1 := t \to \frac{1}{2}\,\frac{-\frac{1}{2}k\,C\,R\,\mathbf{e}^{\left(-\frac{2\,t}{C\,R}\right)} + k\,t + \frac{R\,k\,C}{2}}{C} \qquad u2 := t \to \frac{1}{2}\,\frac{\frac{1}{2}k\,C\,R\,\mathbf{e}^{\left(-\frac{2\,t}{C\,R}\right)} - \frac{R\,k\,C}{2} + k\,t}{C}$$



2. À l'instant $t = T$, on referme l'interrupteur et l'on se propose d'étudier le régime de décharge des condensateurs.

   Écrire un programme en langage MAPLE qui aboutisse au tracé à l'écran des variations de $u_1(t)$ et $u_2(t)$ pour t appartenant à l'intervalle $[T..4T]$

   Note : au moment de la fermeture de l'interrupteur u_1, u_2 et i_2 doivent être continus tandis que i_1 subit nécessairement une discontinuité.

   Déterminer numériquement que le temps au bout duquel le courant i_2 change de sens.

```
> restart:
  i1:=C*diff(u1(t),t):
  i2:=C*diff(u2(t),t):
  u3:=R*i2:
  loi_des_noeuds:=i1+i2=k:
  loi_des_mailles:=u1(t)=u3+u2(t):
  lois_de_Kirchhoff:=loi_des_noeuds,loi_des_mailles:
  tensions:=u1(t),u2(t):
  conditions_initiales:=u2(0)=0,u1(0)=0:
  dsolve({lois_de_Kirchhoff,conditions_initiales},{tensions}):
  assign(%):
  v1:=unapply(u1(t),t):v2:=unapply(u2(t),t):
  R:=500:C:=1000e-6:T:=1:k:=50e-3:

  u1:='u1':u2:='u2':
  loi_des_mailles2:=u1(t)+R*(i1+i2)=0:
  lois_de_Kirchhoff2:=loi_des_mailles,loi_des_mailles2:
  conditions_initiales2:=u2(T)=v2(T),u1(T)=v1(T):
  dsolve({lois_de_Kirchhoff2,conditions_initiales2},{tensions}):
  assign(%):
  u1:=unapply(u1(t),t):u2:=unapply(u2(t),t):

  plot([u1,u2],T..4*T,color=[red,green],thickness=3);
```



3. Assembler les deux programmes précédents pour aboutir au tracé à l'écran, sur un même graphe, des variations de $u_1(t)$, $u_2(t)$ et $Ri_2(t)$ pour t appartenant à l'intervalle $[0..4T]$.

```
> restart:
  i1:=C*diff(u1(t),t):
  i2:=C*diff(u2(t),t):
  u3:=R*i2:
  loi_des_noeuds:=i1+i2=k:
  loi_des_mailles:=u1(t)=u3+u2(t):
  lois_de_Kirchhoff:=loi_des_noeuds,loi_des_mailles:
  tensions:=u1(t),u2(t):
  conditions_initiales:=u2(0)=0,u1(0)=0:
  dsolve({lois_de_Kirchhoff,conditions_initiales},{tensions}):
  assign(%):
  v1:=unapply(u1(t),t):v2:=unapply(u2(t),t):
  #R:=500:C:=1000e-6:T:=1:k:=50e-3:

  u1:='u1':u2:='u2':
  loi_des_mailles2:=u1(t)+R*(i1+i2)=0:
  lois_de_Kirchhoff2:=loi_des_mailles,loi_des_mailles2:
  conditions_initiales2:=u2(T)=v2(T),u1(T)=v1(T):
  dsolve({lois_de_Kirchhoff2,conditions_initiales2},{tensions}):
  assign(%):
  u1:=unapply(simplify(u1(t)),t):u2:=unapply(u2(t),t):
  w1:=proc(t) global T,u1,v1:
      if t<T then v1(t) else u1(t) fi end:
  w2:=proc(t) global T,u2,v2:
      if t<T then v2(t) else u2(t) fi end:
  R:=500:C:=1000e-6:T:=1:k:=50e-3:
  plot([w1,w2,w1-w2],0..4*T,color=[red,green,blue],thickness=3);
```